Geophys. J. Int. (1991) **106**, 709–716

RESEARCH NOTE

3-D rotation of double-couple earthquake sources

Y. Y. Kagan

Institute of Geophysics and Planetary Physics, University of California, Los Angeles, California 90024, *USA*

Accepted 1991 March 22. Received 1991 February 19; in original form 1990 June 20

SUMMARY

We discuss 3-D rotations by which one double-couple earthquake source can be rotated into another arbitrary double-couple. Due to the symmetry of double-couple sources, there are four such rotations. An algorithm is obtained in analytical form which is also available as a computer program solving the inverse problem of 3-D rotation of double-couple earthquake sources, i.e., for each pair of focal mechanisms or seismic moment tensor solutions the program finds all four rotations which rotate one mechanism into another. This algorithm may be used in a wide variety of studies of stress field causing earthquakes, investigations of the relationship between the focal mechanisms and the tectonic features of a seismogenic region, etc. The same inversion algorithm can be used to study the 3-D rotation of any symmetric second-rank tensor, such as the stress or strain tensor.

Key words: double-couple, earthquake source rotation, normalized quaternions.

1 INTRODUCTION

Earthquake focal mechanisms depend both on the ambient stress field, and on local variations in its strength and elastic properties of rocks. Earthquakes strongly perturb both the stress and mechanical properties, often causing fault planes to deviate or 'splay' into branch faults. This branching is essential to the triggering of later earthquakes, and to comprehension of the observed distribution of deep and surface faults. In our previous investigations (Kagan & Knopoff 1985; Kagan 1990, and references therein) we found that the fracture surface of an earthquake is not completely planar, although it can be approximated by a plane, especially in the early stages of rupture. The deviations of rupture surfaces from planes are described by a rotational Cauchy distribution (Kagan 1990). However, the two-point statistical moment of a seismic moment tensor (Kagan & Knopoff 1985) gives only partial information about the degree of non-planarity and 3-D rotation of focal mechanisms. To study branching empirically, we need to derive a statistical distribution for the rotation angles (disorientations) between pairs of earthquake focal mechanisms.

Since a sufficient number of traditional fault-plane and moment tensor solutions is available, we can undertake a study of the correlation of focal mechanisms of individual earthquakes to see whether they yield any information regarding spatial orientations of earthquakes and microearthquakes that make up a fault system. We assume that the fault-plane solutions for individual earthquakes give evidence for the variations in alignment of the respective fracture surfaces and hence in the orientation of portions of an extended fault system. In this paper we discuss 3-D rotations by which one double-couple earthquake source can be rotated into another arbitrary double-couple. A double-couple source is a standard mathematical model for an earthquake focus. Due to the symmetry of double-couple sources there are four such rotations. If one of the rotations is small, we can often ignore the other three rotations; however, even cursory inspection of focal mechanism maps (Goter 1987) shows that often such disorientations are large and hence all four rotations need to be found and studied.

Although studies of regional stress patterns from earthquake focal mechanism data (Michael 1987; Jones 1988; Oppenheimer, Reasenberg & Simpson 1988; Zoback 1989, and references therein) yield a significant insight into the fracture process and its variation in time and space, the standard methods of interpretation depend strongly on some, sometimes arbitrary, assumptions. Moreover, in places where focal mechanisms are strongly disoriented, these methods depend on preliminary regionalization of mechanisms which also leads to some subjectivity of the results. Inverting double-couple solutions for sources of acoustic emission in rock specimens (House *et al.* 1989) seems to indicate that the procedure is unreliable in evaluation of the state of stress.

Giardini & Woodhouse (1984), Frohlich & Willemann (1987), and Michael (1989) studied the clustering of aftershock hypocentres with respect to focal mechanisms of main shocks, as well as the distribution of angles between the median plane of subducting lithosphere and focal planes of earthquakes. These studies may also benefit from ability to determine mutual rotations of focal mechanisms of earthquakes. The study of focal mechanism rotations will allow for a better prediction of the future development of rupture during earthquake sequences which might contribute to better earthquake forecasting. Thus it is important to have an inversion scheme which, if given two earthquake focal mechanisms, yields all of the four rotations of one double-couple to be superimposed upon another. The inversion algorithm described below, is also presented as a FORTRAN program (see Appendix).

2 CALCULATION OF THE ROTATION QUATERNION FOR DOUBLE-COUPLE

Using the known correspondence between normalized quaternions and 3-D rotations (see, for example, Klein 1932; Le Pichon, Francheteau & Bonnin 1973, p. 38 and their appendix; Altmann 1986, chapter 12; Chang, Stock & Molnar 1990; Kagan 1990), we have compiled a computer program to calculate the normalized quaternion corresponding to an arbitrary double-couple. As Altmann (1986) and Chang *et al.* (1990) discuss, the quaternion parametrization of the 3-D rotation has many advantages. These authors also discuss other methods for parametrization of the 3-D rotation, like the Euler angles, Cayley–Klein parameters, etc. The quaternion $\boldsymbol{q}$ is defined as

$$\boldsymbol{q} = q_0 + q_1\boldsymbol{i} + q_2\boldsymbol{j} + q_3\boldsymbol{k}. \tag{1}$$

The first quaternion's component (q_0) is its scalar part, q_1, q_2, and q_3 are components of a 'pure' quaternion; the imaginary units $\boldsymbol{i}$, $\boldsymbol{j}$, and $\boldsymbol{k}$ obey the following multiplication rules:

$$\boldsymbol{i}^2 = \boldsymbol{j}^2 = \boldsymbol{k}^2 = -1, \qquad \boldsymbol{ij} = -\boldsymbol{ji} = \boldsymbol{k}, \qquad \boldsymbol{ki} = -\boldsymbol{ik} = \boldsymbol{j}, \qquad \boldsymbol{jk} = -\boldsymbol{kj} = \boldsymbol{i}. \tag{2}$$

From (2) it is seen that the multiplication of quaternions is not commutative, i.e., depends on the order of multiplicands, the non-commutability also a property of finite 3-D rotations. The conjugate $\boldsymbol{q}^*$ and inverse $\boldsymbol{q}^{-1}$ of a quaternion are defined as

$$\boldsymbol{q}^* = q_0 - q_1\boldsymbol{i} - q_2\boldsymbol{j} - q_3\boldsymbol{k}, \qquad \boldsymbol{qq}^{-1} = 1. \tag{3}$$

We take a normalized quaternion as (1), where

$$q_0^2 + q_1^2 + q_2^2 + q_3^2 = 1. \tag{4}$$

The normalized quaternion defines a 3-D rotation, i.e., the rotation angle is determined as $\Phi = 2\arccos(q_0)$. The vector part of a quaternion corresponds to the rotation axis (Altmann 1986). For the normalized quaternion

$$\boldsymbol{q}^* = \boldsymbol{q}^{-1}. \tag{5}$$

Using normalized quaternions we calculate rotated vector $\mathbf{R}(\boldsymbol{v})$ by using rules of quaternion multiplication (2):

$$\mathbf{R}(\boldsymbol{v}) = \boldsymbol{qvq}^{-1}. \tag{6}$$

Since a double-couple focal mechanism is characterized by three degrees of freedom, we can obtain an appropriate correspondence of the double-couple source with normalized quaternions. In particular, the quaternion $\boldsymbol{1} = [1, 0, 0, 0]$ is taken to correspond to the double-couple with the T axis $(0, 0)$ and the P axis $(0, 90)$ which we call the 'original' (non-rotated) position of a double-couple. The first value in parentheses is the plunge angle in degrees, the second value is the azimuth. The original, right-handed system of source coordinates consists of the T axis pointing north, the P axis pointing east, and the B axis pointing down. It is easy to see that only four right-handed coordinate systems can be formed from these three axes.

We consider an earthquake focal mechanism to be represented in two fashions: (a) through plunge (β) and azimuth (α) of the T and P axes; and (b) through fault plane angles—λ (slip), δ (dip), and dip direction ϕ (azimuth) (Aki & Richards 1980, fig. 4.13; Ben-Menahem & Singh 1981, fig. 4.26). In the first case, we calculate components of the T axis as follows:

$$t_x = \cos(\alpha)\cos(\beta), \qquad t_y = \sin(\alpha)\cos(\beta), \qquad t_z = \sin(\beta), \tag{7}$$

where $\mathbf{t}$ is a unit vector in the direction of the T axis. The components of the P axis are calculated in a similar manner.

In the second case (b), we calculate components of the slip $\mathbf{u}$ and fault normal $\boldsymbol{v}$ vectors (Aki & Richards 1980, equation 4.83; Ben-Menahem & Singh 1981, equation 4.122):

$$u_x = \cos(\lambda)\sin(\phi) - \sin(\lambda)\cos(\delta)\cos(\phi), \qquad u_y = -\cos(\lambda)\cos(\phi) - \sin(\lambda)\cos(\delta)\sin(\phi), \qquad u_z = -\sin(\lambda)\sin(\delta), \tag{8}$$

and

$$v_x = \sin(\delta)\cos(\phi), \qquad v_y = \sin(\delta)\sin(\phi), \qquad v_z = -\cos(\delta). \tag{9}$$

t and **p** unit vectors are obtained as $\mathbf{t} = (\boldsymbol{v} + \mathbf{u})/\sqrt{2}$ and $\mathbf{p} = (\boldsymbol{v} - \mathbf{u})/\sqrt{2}$; to ensure orthogonality of all the three axes and proper 'handedness' of the coordinate system formed by the T, P, and B axes, the null unit vector **b** is computed as a vector product of **t** and **p** for both cases (a) and (b).

The T, P, and B axes specify a rotated system of coordinates for the source, **R**. We use the known correspondence between the orthogonal matrix and the normalized quaternion (Moran 1975, equation 6; Altmann 1986, p. 162)

$$\mathbf{R} = \begin{vmatrix} t_1 & p_1 & b_1 \\ t_2 & p_2 & b_2 \\ t_3 & p_3 & b_3 \end{vmatrix} = \begin{vmatrix} q_0^2 + q_1^2 - q_2^2 - q_3^2 & 2(q_1 q_2 - q_0 q_3) & 2(q_1 q_3 + q_0 q_2) \\ 2(q_1 q_2 + q_0 q_3) & q_0^2 - q_1^2 + q_2^2 - q_3^2 & 2(q_2 q_3 - q_0 q_1) \\ 2(q_1 q_3 - q_0 q_2) & 2(q_2 q_3 + q_0 q_1) & q_0^2 - q_1^2 - q_2^2 + q_3^2 \end{vmatrix} \tag{10}$$

to obtain the quaternion's components. The above formula may be obtained by applying (6) to each of original **t**, **p**, and **b** vectors. For example, if q_0 is not close to zero

$$q_0 = \tfrac{1}{2}(t_1 + p_2 + b_3 + 1)^{1/2}, \qquad q_1 = (p_3 - b_2)/(4q_0), \qquad q_2 = (b_1 - t_3)/(4q_0), \qquad q_3 = (t_2 - p_1)/(4q_0). \tag{11}$$

Since as many as three of the quaternion components may be close to zero, it is computationally simpler to choose the component with a maximum absolute value and use it to calculate the three other components. The formulae which are similar to (11) can be easily derived from (10).

The normalized quaternion found in (11) corresponds to the rotation of a coordinate system connected with a double-couple source from initial position into an arbitrary position. Since a clockwise rotation is equivalent to a counterclockwise rotation about the same axis viewed from the opposite direction, to make the problem unique, we use only counterclockwise rotations corresponding to positive angles of rotation (Altmann 1986, p. 152) with a rotation pole distributed over the whole sphere. As a measure of the disorientation, we use the value of the rotation angle Φ which is necessary for rotating the focal mechanism from one position into another ($0 \le \Phi \le 180°$). This angle depends on the degree of initial disorientation and on the symmetry properties of the source. A double-couple source has the symmetry of a rectangular box with unequal sides. The symmetries of the double-couple make the rotation of a source non-unique. Therefore rotation (11) is not necessarily a minimum rotation, i.e., with a minimum angle Φ.

The double-couple focal mechanism can be rotated from one position into another by four different rotations (Kagan 1990), thus $\boldsymbol{q}$ in (11) corresponds to one of these rotations. To find three other rotations we multiply the normalized quaternion (11) by $\pm\boldsymbol{i}$ or $\pm\boldsymbol{j}$, or $\pm\boldsymbol{k}$ (which are transformations of the quaternion group, see Mermin 1979, pp. 618–619; Altmann 1986, p. 150); e.g.

$$\boldsymbol{q}' = \boldsymbol{q}\boldsymbol{i}, \tag{12}$$

where the quaternion $\boldsymbol{i}$ is $\boldsymbol{i} = [0, 1, 0, 0]$. As a result of these multiplications, the quaternion components are permuted and change their sign. Since quaternions of the opposite sign correspond to the same rotation, we change the quaternion's sign so that its scalar part is positive, corresponding to the positive value of Φ (i.e., the counterclockwise rotation). The end result of all of these four rotations is the same focal mechanism. Then we may choose the rotation which has the smallest rotation angle among the four rotations obtained.

Therefore, to find the minimum rotation of a double-couple, we replace the quaternion's scalar component by the largest (in absolute value) of all of the components available, $q_{\max}$, and then calculate the rotation angle $\Phi_{\min} = 2 \arccos(q_{\max})$. Since the largest of the four components of a normalized quaternion cannot be smaller than 0.5, the minimum rotation angle cannot exceed 120° (Kagan 1990).

3 3-D ROTATION OF DOUBLE-COUPLES

In the previous section we considered the rotation of a double-couple source from its original position. The rotation from one arbitrary position into another is more complicated. As an example let us consider two solutions for earthquakes in the Southern Pacific on 1986 June 5 and June 24 as given by the HARVARD catalogue of the seismic moment tensor inversions (Dziewonski *et al.* 1990, and references therein). The orientation of the T axis for the first earthquake is (0, 226) and the P axis is (0, 136). The second event has (0, 233) and (0, 143), respectively. The four available counterclockwise rotations by which the first focal mechanism can be superimposed over the second solution, can be found by inspection. These rotations are: 7° about the vertical axis looking from below, 173° about the vertical axis looking from above, and two rotations of 180° each, about the bisector of the angles between the corresponding T and P axes for both earthquakes. If one of the rotation angles is small, this rotation can be usually found relatively easily by trial and error. Moreover, averaging the positions of the T, P, and B axes of several focal mechanisms on a reference sphere, produces reasonably good estimates of an average mechanism and its variations for small rotations.

However, in the case of large rotations, we need to find all of the four rotations to be able to choose a more appropriate one. Moreover, straightforward averaging of the axes' positions becomes more questionable when the rotation angles approach 90°, and we need to choose which of the two positions of any axis on the reference sphere is to be used. If we consider as an example two solutions for the earthquakes in New Guinea which occurred on 1977 January 6 and 1980 September 26 (Dziewonski *et al.* 1990), the rotations between these solutions are not so obvious. The T axis for the first solution has (24, 120), the values for the P axis are (41, 232). For the second focal mechanism these values are (55, 295) for the T axis and (17, 51) for the P axis.

Suppose we want to determine all of the possible rotations from one solution $\pm \boldsymbol{q}_1$ into another solution $\pm \boldsymbol{q}_2$,

$$\boldsymbol{q}_2 = \boldsymbol{q}'\boldsymbol{q}_1, \tag{13}$$

where $\boldsymbol{q}'$ is a quaternion corresponding to one of the rotations, transforming $\boldsymbol{q}_1$ into $\boldsymbol{q}_2$. In terms of composition of rotations, (13) assumes that the original quaternion $[1, 0, 0, 0]$ is firstly rotated by $\boldsymbol{q}_1$, then by $\boldsymbol{q}'$ to obtain $\boldsymbol{q}_2$. To determine $\boldsymbol{q}'$ we write

$$\boldsymbol{q}' = \boldsymbol{q}_2\boldsymbol{q}_1^{-1}, \tag{14}$$

see equations (3) and (5). To find three other solutions we multiply $\boldsymbol{q}_1$ or $\boldsymbol{q}_2$ by $\boldsymbol{i}$, $\boldsymbol{j}$, $\boldsymbol{k}$, and repeat the calculations. Out of 16 possible combinations in (14), only four yield different resulting quaternions. It can be shown that, alternatively, all these solutions can be obtained through

$$\boldsymbol{q}'' = \boldsymbol{a}\boldsymbol{q}', \tag{15}$$

where

$$\boldsymbol{a} = \boldsymbol{q}_2\boldsymbol{b}\boldsymbol{q}_2^{-1}, \tag{16}$$

and $\boldsymbol{b}$ is either $\boldsymbol{i}$, or $\boldsymbol{j}$, or $\boldsymbol{k}$. We obtain a solution corresponding to the minimum rotation angle by choosing a quaternion in (15) with a maximum scalar part, q_0. If $\boldsymbol{q}_1 = [1, 0, 0, 0]$ in (13), $\boldsymbol{q}' = \boldsymbol{q}_2$, then we obtain

$$\boldsymbol{q}'' = \boldsymbol{a}\boldsymbol{q}' = \boldsymbol{q}_2\boldsymbol{b}\boldsymbol{q}_2^{-1}\boldsymbol{q}_2 = \boldsymbol{q}'\boldsymbol{b}, \tag{17}$$

as discussed in Section 2 (see equation 12).

The value of the rotation angle Φ and the spherical coordinates, θ and ψ, of the rotation axis on a reference sphere are then calculated (Moran 1975; Altmann 1986, p. 223)

$$\Phi = 2\arccos(q_0), \qquad \theta = \arccos[q_3/\sin(\Phi/2)], \qquad \psi = \arctan(q_2/q_1), \quad \text{if} \quad \psi \le 0, \quad \text{then} \quad \psi = 360° + \psi, \tag{18}$$

where ψ is an azimuth ($0 \le \psi < 360°$) and θ is a colatitude ($0 \le \theta \le 180°$); $\theta = 0$ corresponds to the vector pointing down. For two focal double-couples in New Guinea discussed above, we obtain the following values of quaternions in (13): for the first focal mechanism

$$\boldsymbol{q}_1 = [0.355, 0.233, 0.820, 0.383], \tag{19a}$$

or, using (12)

$$\boldsymbol{q}_1 = [-0.233, 0.355, -0.383, 0.820], \tag{19b}$$

and for the second earthquake

$$\boldsymbol{q}_2 = [-0.041, -0.502, -0.356, 0.787]. \tag{20}$$

The quaternion corresponding to the minimum rotation angle is

$$\boldsymbol{q}_{\min} = [0.696, 0.322, -0.152, 0.624]. \tag{21}$$

The four possible rotation angles are 102.8°, 104.3°, 124.1°, and 165.9°; the spherical coordinates of the rotation poles are (24.8, 101.2), (257.5, 79.7), (144.8, 105.2), and (96.8, 16.7), respectively, where the first value in parentheses is an azimuth in degrees and the second number is a colatitude angle. The FORTRAN program listed in the Appendix, contains several more examples of rotation determination.

4 DISCUSSION

How can we use the proposed inversion scheme for a double-couple mechanism rotation? Depending on the problem at hand, we can use either the minimum rotation for a study of disorientation of earthquake focal mechanisms, or, for instance, we can use the rotation about the axis closest to the normal to the assumed fault plane. Elsewhere, we study distributions of these angles as well as distributions of rotation poles on a reference sphere. Previously, we have calculated a distribution of rotation angles for a completely random 3-D rotation of a double-couple source (Kagan 1990). This distribution may be compared to a distribution of disorientation angles between actual fault planes or between seismic moment tensor solutions.

The general stress tensor, i.e., a symmetric tensor with unequal principal stresses, has the same symmetry properties as a double-couple. Therefore, an inversion algorithm for the rotation of a double-couple source can be used for obtaining a 3-D rotation of practically any symmetric second-rank tensor.

In Section 2 we have considered the computation of a normalized quaternion describing the rotation of a double-couple from its initial position into an arbitrary position. It is also shown that the quaternions are also a concise alternative representation of earthquake double-couple sources. Therefore, this representation may be used to infer average regional focal mechanisms for various faults, their variations and uncertainties (Chang *et al.* 1990), as well as a distribution of rotations of focal solutions in branching faults and other properties which do not depend on pairwise correlation of double-couples. A study of focal mechanism rotations before strong earthquakes could be conducted with a hope of finding precursory phenomena.

We note that the problems discussed above, are similar to those of the maximum likelihood estimate of the parameters of 3-D rotations that have been considered by Moran (1975), and by Thompson & Prentice (1987). Prentice (1987) discusses more complicated problems of fitting a smooth (interpolated) rotation path to a set of rotation data, or finding an average rotation for a set of matched pairs of rotation matrices (Prentice 1989). The problem of focal mechanism rotations is more difficult than the studies above because of the symmetry properties of a double-couple. However, since the above studies use the quaternion representations of the rotations as an input to statistical techniques, their results may be used in analysis of earthquake fault branching and its non-planarity.

ACKNOWLEDGMENTS

This research was supported in part by Grant EAR 88-04883 of the National Science Foundation. The author is indebted to D. D. Jackson of UCLA for his valuable discussions. Publication Number 3453, Institute of Geophysics and Planetary Physics, University of California, Los Angeles, CA 90024.

REFERENCES

Aki, K. & Richards, P., 1980. *Quantitative Seismology,* Freeman, San Francisco.

Altmann, S. L., 1986. *Rotations, Quaternions and Double Groups,* Clarendon Press, Oxford.

Ben-Menahem, A. & Singh, S. J., 1981. *Seismic Waves and Sources,* Springer-Verlag, New York.

Chang, T., Stock, J. & Molnar, P., 1990. The rotation group in plate tectonics and the representation of uncertainties of plate reconstruction, *Geophys. J. Int.,* **101,** 649–661.

Dziewonski, A. M., Ekstrom, G., Woodhouse, J. H. & Zwart, G., 1990. Centroid-moment tensor solutions for October–December 1989, *Phys. Earth planet. Inter.,* **62,** 194–207.

Frohlich, C. & Willemann, R. J., 1987. Statistical methods for comparing directions to the orientations of focal mechanisms and Wadati–Benioff zone, *Bull. seism. Soc. Am.,* **77,** 2135–2142.

Giardini, D. & Woodhouse, J. H., 1984. Deep seismicity and modes of deformation in Tonga subduction zone, *Nature,* **307,** 505–509.

Goter, S. K., 1987. Global distribution of first-motion focal mechanisms, 1981–1985, *World Map,* US Geol. Surv., Denver, CO.

House, L. S., Kranz, R. L., Nishizawa, O. & Satoh, T., 1989. Use of acoustic emissions to investigate stresses in laboratory rock samples, *EOS, Trans. Am. geophys. Un.,* **70,** 1340.

Jones, L. M., 1988. Focal mechanisms and state of stress on the San Andreas fault in southern California, *J. geophys. Res.,* **93,** 8869–8891.

Kagan, Y. Y., 1990. Random stress and earthquake statistics: spatial dependence, *Geophys. J. Int.,* **102,** 573–583.

Kagan, Y. Y. & Knopoff, L., 1985. The two-point correlation function of the seismic moment tensor, *Geophys. J. R. astr. Soc.,* **83,** 637–656.

Klein, F., 1932. *Elementary Mathematics from an Advanced Standpoint. Vol. I. Arithmetic, Algebra, Analysis,* Macmillan, London.

Le Pichon, X., Francheteau, J. & Bonnin, J., 1973. *Plate Tectonics,* Elsevier, London.

Mermin, N. D., 1979. The topological theory of defects in ordered media, *Rev. Mod. Phys.,* **51,** 591–648.

Michael, A. J., 1987. Use of focal mechanisms to determine stress: a control study, *J. geophys. Res.,* **92,** 357–368.

Michael, A. J., 1989. Spatial patterns of aftershocks of shallow earthquakes in California and implications for deep focus earthquakes, *J. geophys. Res.,* **94,** 5615–5626.

Moran, P. A. P., 1975. Quaternions, Haar measure and estimation of paleomagnetic rotation, in *Perspectives in Probability and Statistics,* pp. 295–301, ed. Gani, J., Academic Press, London.

Oppenheimer, D. H., Reasenberg, P. A. & Simpson, R. W., 1988. Fault plane solutions for the 1984 Morgan Hill, California, earthquake sequence: evidence for state of stress on the Calaveras fault, *J. geophys. Res.,* **93,** 9007–9026.

Prentice, M. J., 1987. Fitting smooth path to rotation data, *Appl. Statist.,* **36,** 325–331.

Prentice, M. J., 1989. Spherical regression on matched pairs of orientation statistics, *J. R. Stat. Soc.,* **B51,** 241–248.

Thompson, R. & Prentice, M. J., 1987. An alternative method of calculating finite plate rotations, *Phys. Earth planet. Inter.,* **48,** 78–83.

Zoback, M. L., 1989. State of stress and modern deformation of the northern Basin and Range province, *J. geophys. Res.,* **94,** 7105–7128.

APPENDIX

```
C
C      BEGIN OF THE DCROT PROGRAM
C
C      ROTATION ANGLE AND AXIS DIRECTION CALCULATIONS
C   FOR ALL FOUR POSSIBLE ROTATIONS OF DOUBLE-COUPLE SOURCE
C
C          DRIVER PROGRAM -- February 15, 1991
C
C
      IMPLICIT REAL*8 (A-H,O-Z)
C
C
c   1985  2   16   16   28   14.90    39.690  142.690   39       66 264     22 109
c   1987  4    7    0   40   49.50    37.300  141.750   31       61 296     29 114
      INTEGER*2 EQA1(4) /66,264, 22,109/, EQA2(4) /61,296, 29,114/
c
c   1985  7    2   12   34   56.10    40.630  143.900   24       57 300     33 113
c   1987  1    9    6   14   50.00    39.800  141.380   60       53  89     35 289
      INTEGER*2 EQB1(4) /57,300, 33,113/, EQB2(4) /53,89, 35,289/
C
c   1977  1    6    6   11   50.50    -2.990  144.790   11       24 120     41 232
c   1980  9   26   15   20   42.50    -3.000  142.430   25       55 295     17  51
      INTEGER*2 EQC1(4) /24,120, 41,232/, EQC2(4) /55,295, 17,51/
c   1986  6    5    9    1   20.30   -36.290  -97.120   15        0 226      0 136
c   1986  6   24   23   53   32.30   -36.550 -100.450   15        0 233      0 143
      INTEGER*2 EQD1(4) /0,226, 0,136/, EQD2(4) /0,233, 0,143/
c
      INTEGER*2 EQE1(4) /0,0, 0,90/, EQE2(4) /90,0, 0,0/
C
      COMMON /MOM/ RAD, PERP
C
      HPI = DACOS(0.0D0)
      RAD = 90.0D0/HPI
      WRITE (6, 15)
      CALL FPS4R (EQA1, EQA2)
      WRITE (6, 15)
      CALL FPS4R (EQB1, EQB2)
      WRITE (6, 15)
      CALL FPS4R (EQC1, EQC2)
      WRITE (6, 15)
      CALL FPS4R (EQD1, EQD2)
      WRITE (6, 15)
      CALL FPS4R (EQE1, EQE2)
      WRITE (6, 15)
   15 FORMAT ('1')
C
      STOP
      END
C
      SUBROUTINE FPS4R (EQH1, EQH2)
C
      IMPLICIT REAL*8 (A-H,O-Z)
      REAL*8 QUAT(4), QUAT1(4), QUAT2(4), QUATR(4), QUATR1(4),
     1 QUATR2(4), QUATT1(4), QUATT2(4), QUATC1(4)
      REAL*8 Q1A(4), Q2A(4), Q3A(4), Q1B(4), Q2B(4), Q3B(4), Q4(4)
      INTEGER*2 EQH1(4), EQH2(4)
      COMMON /MOM/ RAD, PERP
C
      WRITE (6, 20) EQH1, EQH2
      WRITE (6, 10)
C
      CALL QUATFPS (QUAT1, EQH1, 0)
      CALL SPHCOOR (QUAT1, ANGL, THETA, PHI)
      WRITE (6, 30) ANGL, THETA, PHI, PERP
      ICODE = 0
      CALL BOXTEST (QUAT1, QUATR1, QM, ICODE)
      WRITE (6, 40) QUAT1, QUATR1
      WRITE (6, 50) ICODE, QM
      CALL SPHCOOR (QUATR1, ANGL, THETA, PHI)
      WRITE (6, 30) ANGL, THETA, PHI
      WRITE (6, 10)
C
      CALL QUATFPS (QUAT2, EQH2, 0)
      CALL SPHCOOR (QUAT2, ANGL, THETA, PHI)
      WRITE (6, 30) ANGL, THETA, PHI, PERP
      ICODE = 0
      CALL BOXTEST (QUAT2, QUATR2, QM, ICODE)
      WRITE (6, 40) QUAT2, QUATR2
      WRITE (6, 50) ICODE, QM
      CALL SPHCOOR (QUATR2, ANGL, THETA, PHI)
      WRITE (6, 30) ANGL, THETA, PHI
C
      WRITE (6, 10)
      CALL F4R1 (QUAT1, QUAT2, Q1A, 1)
      CALL F4R2 (QUAT1, QUAT2, Q1B, 1)
      WRITE (6, 10)
      WRITE (6, 40) Q1A, Q1B
C
      WRITE (6, 10)
      CALL F4R1 (QUAT1, QUAT2, Q2A, 2)
      CALL F4R2 (QUAT1, QUAT2, Q2B, 2)
      WRITE (6, 10)
      WRITE (6, 40) Q2A, Q2B
C
      WRITE (6, 10)
      CALL F4R1 (QUAT1, QUAT2, Q3A, 3)
      CALL F4R2 (QUAT1, QUAT2, Q3B, 3)
      WRITE (6, 10)
      WRITE (6, 40) Q3A, Q3B
C
      WRITE (6, 10)
      CALL F4R1 (QUAT1, QUAT2, Q4, 4)
      CALL F4R2 (QUAT1, QUAT2, Q4, 4)

C
      WRITE (6, 10)
      WRITE (6, 40) Q4
      WRITE (6, 10)
C
   10 FORMAT (' ')
   20 FORMAT ('0 EQH1, EQH2 = ', 4I5, 4x, 4I5)
   30 FORMAT ('  ANGL, THETA, PHI = ', 7F14.7)
   40 FORMAT (' ', 9G14.6)
   50 FORMAT (' ICODE, QM = ', I5, 3F14.7)
      RETURN
      END
```

```
C
      SUBROUTINE F4R1 (Q1, Q2, Q, ICODE)
c
c     Q = Q2*(Q1*(I,J,K,1))**(-1)
c
c     Since F4R1 and F4R2 yield the same results, only one subroutine
c     is needed; both programs are kept here for testing purposes.
c
      IMPLICIT REAL*8 (A-H,O-Z)
      REAL*8 Q(4), Q1(4), Q2(4), QR1(4),
     1 QT1(4), QT2(4), QC1(4)
C
      CALL BOXTEST (Q1, QR1, QM, ICODE)
C      WRITE (6, 20) QR1
C
      CALL QUATD (QR1, Q2, Q)
C      WRITE (6, 20) Q1, Q2
C      CALL QUATP (Q1, Q, QT2)
C      WRITE (6, 20) QT2, Q
C
      CALL SPHCOOR (Q, ANGL, THETA, PHI)
C
      WRITE (6, 10) ANGL, THETA, PHI
C      WRITE (6, 20) Q, QR, QM
   10 FORMAT ('  ANGL, THETA, PHI = ', 3F14.7)
   20 FORMAT (' ', 9G14.6)
      RETURN
      END


C
      SUBROUTINE F4R2 (Q1, Q2, Q, ICODE)
c
c     Q = (Q2*(I,J,K,1))*Q1**(-1)
c
      IMPLICIT REAL*8 (A-H,O-Z)
      REAL*8 Q(4), Q1(4), Q2(4), QR2(4),
     1 QT1(4), QT2(4), QC1(4)
C
      CALL BOXTEST (Q2, QR2, QM, ICODE)
C      WRITE (6, 20) QR2
C
      CALL QUATD (Q1, QR2, Q)
C      WRITE (6, 20) Q1, Q2
C      CALL QUATP (Q1, Q, QT2)
C      WRITE (6, 20) QT2, Q
C
      CALL SPHCOOR (Q, ANGL, THETA, PHI)
C
      WRITE (6, 10) ANGL, THETA, PHI
C      WRITE (6, 20) Q, QR, QM
   10 FORMAT ('  ANGL, THETA, PHI = ', 3F14.7)
   20 FORMAT (' ', 9G14.6)
      RETURN
      END

C
      SUBROUTINE BOXTEST (Q1, Q2, QM, ICODE)
c
c     for ICODE=0 finds minimal rotation quaternion
c     for ICODE=N finds rotation quaternion Q2 = Q1*(i,j,k,1),
c     for N=1,2,3,4
c
      IMPLICIT REAL*8 (A-H, O-Z)
      REAL*8 Q1(4), Q2(4), QUATT(4)
      REAL*8 QUAT(4, 3) /1.0D0, 0.0D0, 0.0D0, 0.0D0,
     2                   0.0D0, 1.0D0, 0.0D0, 0.0D0,
     3                   0.0D0, 0.0D0, 1.0D0, 0.0D0/
C
      IF (ICODE.NE.0) GO TO 15
      ICODE = 1
      QM = DABS(Q1(1))
      DO 10 IXC = 2, 4
      IF (QM.GE.DABS(Q1(IXC))) GO TO 10
      QM = DABS(Q1(IXC))
      ICODE = IXC
   10 CONTINUE
   15 CONTINUE
C
      DO 20 IXC = 1, 4
      Q2(IXC) = Q1(IXC)
   20 CONTINUE
C
      IF (ICODE.EQ.4) GO TO 40
      DO 30 IXC = 1, 4
      QUATT(IXC) = QUAT(IXC, ICODE)
   30 CONTINUE
      CALL QUATP (QUATT, Q1, Q2)
   40 CONTINUE
C
      IF (Q2(4).GT.0.0D0) GO TO 60
      DO 50 IXC = 1, 4
      Q2(IXC) = - Q2(IXC)
   50 CONTINUE
   60 CONTINUE
      QM = Q2(4)
C
      RETURN
      END



C
      SUBROUTINE SPHCOOR (QUAT, ANGL, THETA, PHI)
c
c     for the rotation quaternion QUAT the subroutine finds the
c     rotation angle (ANGL) of a counterclockwise rotation and
c     spherical coordinates (colatitude THETA, and azimuth PHI) of the
c     rotation pole (intersection of the axis with reference sphere);
c     THETA=0 corresponds to the vector pointing down.
c
      IMPLICIT REAL*8 (A-H,O-Z)
      REAL*8 QUAT(4)
C
      IF (QUAT(4).LT.0.0D0) THEN
      DO 10 ISM = 1, 4
      QUAT(ISM) = - QUAT(ISM)
   10 CONTINUE
      END IF
```

```
C
      Q4N = DSQRT(1.0D0 - QUAT(4)**2)
      COSTH = 1.0
      IF (DABS(Q4N).GT.1.0D-10) COSTH = QUAT(3)/Q4N
      IF (DABS(COSTH).GT.1.0) COSTH = JIDINT(COSTH)
      THETA = DACOSD(COSTH)
      ANGL = 2.0D0*DACOSD(QUAT(4))
      PHI = 0.0D0
      IF (DABS(QUAT(1)).GT.1.0D-10.OR.DABS(QUAT(2)).GT.1.0D-10)
     1 PHI = DATAN2D(QUAT(2),QUAT(1))
      IF (PHI.LT.0.0D0) PHI = PHI + 360.0D0
C
      RETURN
      END



C
      SUBROUTINE QUATFPS (QUAT, EQH, ICODE)
      IMPLICIT REAL*8 (A-H, O-Z)
      REAL*8 QUAT(4)
      INTEGER*2 EQH(4)
      COMMON /MOM/ RAD, PERP
C
C     THIS ROUTINE CALCULATES ROTATION QUATERNION CORRESPONDING TO
C                    EARTHQUAKE FOCAL MECHANISM
C
c     icode=0 -- four input data:  plunge and azimuth of T-axis
c                                  plunge and azimuth of P-axis
c     Since plunge and azimuth of 2 axes are redundant for calculation,
c     (four degrees of freedom vs three degrees that are necessary)
c     and have low accuracy (integer angular degrees), we calculate
c     plane normal (V) and slip vector (S) axes, in order that all axes
c     be orthogonal.
c
c     icode=1 -- three input data: slip angle (SA), dip angle (DA),
c                                  dip direction (DD)
c
c     PERP variable checks orthogonality
c     of T- and P-axes, it should be small (<0.01 or so).
C
      ERR = 1.0D-15
      IC = 1
      IF (ICODE.EQ.1) GO TO 200
      PLG T AX = EQH(1)
      AZM T AX = EQH(2)
      PLG P AX = EQH(3)
      AZM P AX = EQH(4)
C
      T1 = DCOS(AZM T AX/RAD)*DCOS(PLG T AX/RAD)
      T2 = DSIN(AZM T AX/RAD)*DCOS(PLG T AX/RAD)
      T3 = DSIN(PLG T AX/RAD)
      P1 = DCOS(AZM P AX/RAD)*DCOS(PLG P AX/RAD)
      P2 = DSIN(AZM P AX/RAD)*DCOS(PLG P AX/RAD)
      P3 = DSIN(PLG P AX/RAD)
      PERP = T1*P1 + T2*P2 + T3*P3
      IF (PERP.GT.2.0D-02) THEN
      WRITE (6, 140) EQH, T1, T2, T3, P1, P2, P3, PERP
  140 FORMAT (' **** T- AND P-AXES ARE NOT ORTHOGONAL:
     1 EQH, T1, T2, T3, P1, P2, P3, PERP = ', /, 4I4, 7G14.5)
      STOP 35
      END IF
C
      V1 = T1 + P1
      V2 = T2 + P2
      V3 = T3 + P3
      S1 = T1 - P1
      S2 = T2 - P2
      S3 = T3 - P3
      ANORMV = DSQRT(V1*V1 + V2*V2 + V3*V3)
      V1 = V1/ANORMV
      V2 = V2/ANORMV


      V3 = V3/ANORMV
      ANORMS = DSQRT(S1*S1 + S2*S2 + S3*S3)
      S1 = S1/ANORMS
      S2 = S2/ANORMS
      S3 = S3/ANORMS
C
      GO TO 250
  200 CONTINUE
C
      DD = EQH(1)
      DA = EQH(2)
      SA = EQH(3)
      DD = DD/RAD
      DA = DA/RAD
      SA = SA/RAD
      CDD = DCOS(DD)
      SDD = DSIN(DD)
      CDA = DCOS(DA)
      SDA = DSIN(DA)
      CSA = DCOS(SA)
      SSA = DSIN(SA)
      S1 = CSA*SDD - SSA*CDA*CDD
      S2 = - CSA*CDD - SSA*CDA*SDD
      S3 = - SSA*SDA
      V1 = SDA*CDD
      V2 = SDA*SDD
      V3 = - CDA
C
  250 CONTINUE
      AN1 = s2*v3 - v2*s3
      AN2 = v1*s3 - s1*v3
      AN3 = s1*v2 - v1*s2
C      SINV3 = S1*V2*AN3 + S2*V3*AN1 + V1*AN2*S3 -
C     1 S3*V2*AN1 - S1*AN2*V3 - AN3*V1*S2
      D2 = 1.0D0/DSQRT(2.0D0)
      T1 = (V1 + S1)*D2
      T2 = (V2 + S2)*D2
      T3 = (V3 + S3)*D2
      P1 = (V1 - S1)*D2
      P2 = (V2 - S2)*D2
      P3 = (V3 - S3)*D2
      WRITE (6, 100) T1, T2, T3, P1, P2, P3, AN1, AN2, AN3
  100 FORMAT (' T1, T2, T3, P1, P2, P3, AN1, AN2, AN3 = ', /, 9G13.4)
C
      U0 = (T1 + P2 + AN3 + 1.0D0)/4.0D0
      U1 = (T1 - P2 - AN3 + 1.0D0)/4.0D0
      U2 = (-T1 + P2 - AN3 + 1.0D0)/4.0D0
      U3 = (-T1 - P2 + AN3 + 1.0D0)/4.0D0
      UM = DMAX1(U0, U1, U2, U3)
      IF (UM.EQ.U0) GO TO 10
      IF (UM.EQ.U1) GO TO 20
      IF (UM.EQ.U2) GO TO 30
      IF (UM.EQ.U3) GO TO 40
      WRITE (6, 150)
C


   10 CONTINUE
      ICOD = 1*IC
      U0 = DSQRT(U0)
      U3 = (T2 - P1)/(4.0D0*U0)
      U2 = (AN1 - T3)/(4.0D0*U0)
      U1 = (P3 - AN2)/(4.0D0*U0)
      GO TO 50
   20 CONTINUE
      ICOD = 2*IC
      U1 = DSQRT(U1)
      U2 = (T2 + P1)/(4.0D0*U1)
      U3 = (T3 + AN1)/(4.0D0*U1)
      U0 = (P3 - AN2)/(4.0D0*U1)
      GO TO 50
   30 CONTINUE
      ICOD = 3*IC
      U2 = DSQRT(U2)
      U1 = (T2 + P1)/(4.0D0*U2)
      U0 = (AN1 - T3)/(4.0D0*U2)
      U3 = (P3 + AN2)/(4.0D0*U2)
      GO TO 50
   40 CONTINUE
      ICOD = 4*IC
      U3 = DSQRT(U3)
      U0 = (T2 - P1)/(4.0D0*U3)
      U1 = (T3 + AN1)/(4.0D0*U3)
      U2 = (P3 + AN2)/(4.0D0*U3)
   50 CONTINUE
      TEMP = U0*U0 + U1*U1 + U2*U2 + U3*U3
C
      IF (DABS(TEMP - 1.0D0).GT.ERR) THEN
      WRITE (6, 150)
  150 FORMAT (' **** ERROR *****')
      WRITE (6, 90) T1, T2, T3, P1, P2, P3
   90 FORMAT (' T1, T2, T3, P1, P2, P3 = ', /, 6G18.9)
      WRITE (6, 80) AN1, AN2, AN3
   80 FORMAT (' AN1, AN2, AN3 = ', 3G18.9)
      WRITE (6, 120) TEMP, U1, U2, U3, U0
  120 FORMAT (' TEMP, U1, U2, U3, U0 = ', 5G18.9)
      END IF
      QUAT(1) = U1
      QUAT(2) = U2
      QUAT(3) = U3
      QUAT(4) = U0
      WRITE (6, 130) QUAT, ICOD
  130 FORMAT (' QUAT = ', 4G18.9, ' ICOD =', I5)
C
      RETURN
      END



C
      SUBROUTINE QUATP (Q1, Q2, Q3)
C
C  Calculates product of two quaternions Q3 = Q2*Q1,
C  see F. Klein v.1 p.61, or Altmann, 1986, p.156,
C  or Biedenharn and Louck, 1981, p. 185.
c
C  Quaternion is taken here  --  q1*i + q2*j + q3*k + q4
C
      IMPLICIT REAL*8 (A-H, O-Z)
C
      REAL*8 Q1(4), Q2(4), Q3(4)
C
      Q3(1) =  Q1(4)*Q2(1) + Q1(3)*Q2(2) - Q1(2)*Q2(3) + Q1(1)*Q2(4)
      Q3(2) = -Q1(3)*Q2(1) + Q1(4)*Q2(2) + Q1(1)*Q2(3) + Q1(2)*Q2(4)
      Q3(3) =  Q1(2)*Q2(1) - Q1(1)*Q2(2) + Q1(4)*Q2(3) + Q1(3)*Q2(4)
      Q3(4) = -Q1(1)*Q2(1) - Q1(2)*Q2(2) - Q1(3)*Q2(3) + Q1(4)*Q2(4)
      RETURN
      END
C
C
C
C
      SUBROUTINE QUATD (Q1, Q2, Q3)
      IMPLICIT REAL*8 (A-H, O-Z)
C
C     Quaternion division Q3 = Q2*(Q1)**(-1), or Q2 = Q3*Q1
C
      REAL*8 Q1(4), QC1(4), Q2(4), Q3(4)
C
      DO 10 I = 1, 3
      QC1(I) = - Q1(I)
   10 CONTINUE
      QC1(4) = Q1(4)
      CALL QUATP (QC1, Q2, Q3)
C
      RETURN
      END
C
C      END OF THE DCROT PROGRAM
C
```

EXAMPLE OF PROGRAM'S OUTPUT

```
 EQH1, EQH2 =    66  264   22  109       61  296   29  114

T1, T2, T3, P1, P2, P3, AN1, AN2, AN3 =
 -0.4245E-01  -0.4047      0.9135      -0.3019      0.8768      0.3744      -0.9524      -0.2598      -0.1594
QUAT =   -0.245035828      0.720864950       0.397338210E-01  -0.647095349     ICOD =    3
 ANGL, THETA, PHI =     99.3540801    92.9873884   288.7738827     0.0004337
  0.245036     -0.720865     -0.397338E-01  0.647095      0.397338E-01  0.647095      0.245036      0.720865
ICODE, QM =     2     0.7208649
 ANGL, THETA, PHI =     87.7481230    69.2954880    86.4862589

T1, T2, T3, P1, P2, P3, AN1, AN2, AN3 =
  0.2126      -0.4358      0.8746      -0.3558      0.7991      0.4847      -0.9101      -0.4142      0.1480E-01
QUAT =    0.315718670      -0.626835392      -0.281272532E-01   0.711763985     ICOD =    1
 ANGL, THETA, PHI =     89.2427603    92.2949331   296.7330601     0.0002583
  0.315719     -0.626835     -0.281273E-01  0.711764      0.315719     -0.626835     -0.281273E-01  0.711764
ICODE, QM =     4     0.7117640
 ANGL, THETA, PHI =     89.2427603    92.2949331   296.7330601

 ANGL, THETA, PHI =    176.0431671   154.1327695   101.2360137
 ANGL, THETA, PHI =    176.0431671   154.1327695   101.2360137

 -0.849603E-01  0.427670     -0.899271      0.345230E-01 -0.849603E-01  0.427670     -0.899271      0.345230E-01

 ANGL, THETA, PHI =    167.0100624   115.2537837   291.8677364
 ANGL, THETA, PHI =    167.0100624   115.2537837   291.8677364

  0.334706     -0.833963     -0.423890      0.113116      0.334706     -0.833963     -0.423890      0.113116

 ANGL, THETA, PHI =    172.6710792    93.8800345   199.5710993
 ANGL, THETA, PHI =    172.6710792    93.8800345   199.5710993

 -0.938145     -0.333525     -0.675293E-01  0.639133E-01 -0.938145     -0.333525     -0.675293E-01  0.639133E-01

 ANGL, THETA, PHI =     15.4515568    51.2886179    76.0649341
 ANGL, THETA, PHI =     15.4515568    51.2886179    76.0649341

  0.252618E-01  0.101811      0.840735E-01  0.990923
```